CocycleOfExtension

In this example we compute the so-called characteristic class of an extension, which is a 1-cocycle:

```
> restart:
> with(JanetOre): with(homalg):
```

Specify the homalg-table of the ring package JanetOre:

```
> RPO := `JanetOre/homalg`;
```

$$RPO := JanetOre/homalg$$

Use the ring package JanetOre as the default ring package:

```
> `homalg/default` := RPO;
```

$$homalg/default := JanetOre/homalg$$

Define the ring $R := \mathbb{Q}(g,c)[t][D,\delta]$, where g, c are independent transcendentals (since we use "constant coefficients", i.e. t does not occur explicitly one can take the ring to be the polynomial ring $R := \mathbb{Q}(g,c)[D,\delta]$, so for this example one can even use the ring package Involutive, since the computation is commutative; recall $[D,\delta] = 0$):

```
> R := [[t,D,delta],[],[weyl(D,t),shift(delta,t)]];
```

$$R := [[t,\, \mathrm{D},\, \delta],\, [],\, [\mathrm{weyl}(\mathrm{D},\, t),\, \mathrm{shift}(\delta,\, t)]]$$

The matrix of relations of the module:

```
> A:=evalm([[D,-D*delta^2,c/g*D^2*delta],[D*delta^2,-D,c/g*D^2*delta]])
;
```

$$A := \begin{bmatrix} \mathrm{D} & -\mathrm{D}\,\delta^2 & \dfrac{c\,\mathrm{D}^2\,\delta}{g} \\ \mathrm{D}\,\delta^2 & -\mathrm{D} & \dfrac{c\,\mathrm{D}^2\,\delta}{g} \end{bmatrix}$$

The module $M := \mathrm{coker}(R^{1\times 2} \xrightarrow{A} R^{1\times 3})$:

```
> M := Cokernel(A,R);
```

$$M := [[[1,\, 0,\, 0] = [1,\, 0,\, 0],\, [0,\, 1,\, 0] = [0,\, 1,\, 0],\, [0,\, 0,\, 1] = [0,\, 0,\, 1]],$$
$$[[\mathrm{D}\,\delta^2 - \mathrm{D},\, \mathrm{D}\,\delta^2 - \mathrm{D},\, 0],\, [\frac{\mathrm{D}\,g}{c},\, -\frac{\mathrm{D}\,\delta^2\,g}{c},\, \delta\,\mathrm{D}^2]],\, \text{“Presentation”},$$
$$3 + 9\,s + 18\,s^2 + s^3\,(\frac{18}{1-s} + \frac{9}{(1-s)^2} + \frac{1}{(1-s)^3}),\, [18,\, 9,\, 1]]$$

The torsion submodule $M' := \mathrm{t}(M)$:

```
> M_ := TorsionSubmodule(M,R);
```

$$M\_ := [[[1,\, 0] = [1,\, 1,\, 0],\, [0,\, 1] = [0,\, -\frac{g\,(1+\delta^2)}{c},\, \mathrm{D}\,\delta]],\, [[\mathrm{D},\, \frac{c\,\mathrm{D}}{g}],\, [0,\, \mathrm{D}\,\delta^2 - \mathrm{D}]],$$
$$\text{“Presentation”},\, 2 + 5\,s + 9\,s^2 + s^3\,(\frac{9}{1-s} + \frac{4}{(1-s)^2}),\, [9,\, 4,\, 0]]$$

The embedding $\mathrm{t}(M) \overset{\alpha_1}{\hookrightarrow} M$:

```
> alpha1 := TorsionSubmoduleEmb(M,R);
```

$$\alpha 1 := \begin{bmatrix} 1 & 1 & 0 \\ 0 & -\dfrac{g\,(1+\delta^2)}{c} & \mathrm{D}\,\delta \end{bmatrix}$$

The torsion free factor $M'' := M/M' = M/\,\mathrm{t}(M)$:

```
> _M := Cokernel(alpha1,M,R);
```

$$\_M := [[[1,0]=[0,1,0],\ [0,1]=[0,0,1]],\ [[-\frac{g\,\delta^2}{c}-\frac{g}{c},\ \mathrm{D}\,\delta]],\ \text{“Presentation”},$$
$$2+6\,s+s^2\,(\frac{6}{1-s}+\frac{4}{(1-s)^2}+\frac{1}{(1-s)^3}),\ [6,\,4,\,1]]$$

The natural epimorphism $M \overset{\alpha_2}{\twoheadrightarrow} M'' = M/M'$:

```
> alpha2 := CokernelEpi(alpha1,M,R);
```

$$\alpha 2 := \begin{bmatrix} -1 & 0 \\ 1 & 0 \\ 0 & 1 \end{bmatrix}$$

By construction one gets a short exact sequence $0 \to M' \xrightarrow{\alpha_1} M \xrightarrow{\alpha_2} M'' \to 0$:

```
> E1 := M_,alpha1,M,alpha2,_M;
```

$$E1 := [[[1,0]=[1,1,0],\ [0,1]=[0,-\frac{g\,(1+\delta^2)}{c},\ \mathrm{D}\,\delta]],\ [[\mathrm{D},\ \frac{c\,\mathrm{D}}{g}],\ [0,\ \mathrm{D}\,\delta^2-\mathrm{D}]],$$
$$\text{“Presentation”},\ 2+5\,s+9\,s^2+s^3\,(\frac{9}{1-s}+\frac{4}{(1-s)^2}),\ [9,\,4,\,0]],\ \alpha 1,\ [$$
$$[[1,0,0]=[1,0,0],\ [0,1,0]=[0,1,0],\ [0,0,1]=[0,0,1]],$$
$$[[\mathrm{D}\,\delta^2-\mathrm{D},\ \mathrm{D}\,\delta^2-\mathrm{D},\ 0],\ [\frac{\mathrm{D}\,g}{c},\ -\frac{\mathrm{D}\,\delta^2\,g}{c},\ \delta\,\mathrm{D}^2]],\ \text{“Presentation”},$$
$$3+9\,s+18\,s^2+s^3\,(\frac{18}{1-s}+\frac{9}{(1-s)^2}+\frac{1}{(1-s)^3}),\ [18,\,9,\,1]],\ \alpha 2,\ [$$
$$[[1,0]=[0,1,0],\ [0,1]=[0,0,1]],\ [[-\frac{g\,\delta^2}{c}-\frac{g}{c},\ \mathrm{D}\,\delta]],\ \text{“Presentation”},$$
$$2+6\,s+s^2\,(\frac{6}{1-s}+\frac{4}{(1-s)^2}+\frac{1}{(1-s)^3}),\ [6,\,4,\,1]]$$

```
> IsShortExactSeq(E1,R,"VERBOSE");
```

true

This short exact sequence is an *extension* of M'' by M'. To an extension of M'' by M' there corresponds a cocycle in $\mathrm{Ext}^1_R(M'', M')$. To describe this correspondence let $P_2 \xrightarrow{d_2} P_1 \xrightarrow{d_1} P_0 \xrightarrow{d_0} M'' \to 0$ be the beginning of a projective resolution of M''. A 1-cocylce is by definition a morphism $\eta : P_1 \to M'$ with $d_2^*(\eta) = d_2\eta = 0$. I.e. η factors over.

$$P_1/\operatorname{im}(d_2) = P_1/\ker(d_1) \overset{d_1}{\cong} \ker(d_0) =: K_1.$$

K_1 is called the first syzygies module of M, which is due to SCHANUEL's Lemma uniquely defined up to projective equivalence. This establishes the well-known isomorphism between the first derived functor

$$\begin{aligned} \mathrm{R}^1\,\mathrm{Hom}(-,M')(M'') &:= \operatorname{def}(\mathrm{Hom}_R(P_0,M') \xrightarrow{d_1^*} \mathrm{Hom}_R(P_1,M') \xrightarrow{d_2^*} \mathrm{Hom}_R(P_2,M')) \\ &= \{\eta : P_1 \to M' \mid 0 = d_2^*(\eta) := d_2\eta\}/\{d_1^*\psi := d_1\psi \mid \psi : P_0 \to M'\} \end{aligned}$$

and the first right *satellite*[1]

$$\begin{aligned} \mathrm{S}^1\,\mathrm{Hom}(-,M')(M'') &:= \operatorname{coker}(\mathrm{Hom}_R(P_0,M') \xrightarrow{d_1^*} \mathrm{Hom}_R(K_1,M')) \\ &= \{\eta : K_1 \to M'\}/\{d_1^*\psi := d_1\psi \mid \psi : P_0 \to M'\} \end{aligned}$$

(cf. [HS97, Section III.2, Prop. IV.5.8, Exercises IV.7.3 and IX.3.1] and [CE99, III.(6a),(6a')]). In words, the first right satellite of $\mathrm{Hom}_R(-,M')$ applied to M'' is the abelian group of all morphisms

[1]The right satellites of a *left exact* functor coincide with the right derived functors (cf. [CE99, Theorem V.6.1] and [HS97, Prop. IV.5.8]).

$K_1 \to M'$, modulo those which factor over P_0 (i.e. which extend to P_0). The diagram (cf. [HS97, Theorem III.2.4])

(Ext)
$$\begin{array}{ccccccccc} 0 & \longrightarrow & K_1 & \xrightarrow{d_1} & P_0 & \longrightarrow & M'' & \longrightarrow & 0 \\ & & \downarrow{\scriptstyle \eta} & & \downarrow{\scriptstyle \kappa} & & \| & & \\ 0 & \longrightarrow & M' & \xrightarrow{\alpha_1} & M & \xrightarrow{\alpha_2} & M'' & \longrightarrow & 0 \end{array}$$

shows how to compute η by lifting the identity $M'' \xrightarrow{\mathrm{id}} M''$ twice. Even for noncommuative rings, the effective computation of the first lift κ works if P_0 is free (cf. [BR, 3.1.3,(1)]) and the second lift works since α_1 is injective (cf. [BR, 3.1.3,(2)]). Conversely, M is in the above diagram the *pushout* of $M' \xleftarrow{\eta} K_1 \xrightarrow{d_1} P_0$ (cf. [HS97, Exercise II.9.2]), i.e. the cokernel of

$$K_1 \xrightarrow{\begin{pmatrix} d_1 & \eta \end{pmatrix}} P_0 \oplus M' \xrightarrow{\begin{pmatrix} \kappa \\ -\alpha_1 \end{pmatrix}} M \to 0.$$

```
> ext :=
Ext(1,_M,M_,R,"var_to_assign_connecting_hom_proc"='ch',"var_to_assign_
proc_to_express_generators_abstractly"='abstract_cocycle');
```

$$ext := [[[1,\, 0] = \begin{bmatrix} 0 & 1 \end{bmatrix},\, [0,\, 1] = \begin{bmatrix} 1 & 0 \end{bmatrix}],\, [[0,\, \mathrm{D}],\, [\mathrm{D},\, 0],\, [0,\, 1+\delta^2],\, [1+\delta^2,\, 0]],$$
$$\text{“Presentation”},\, 2 + 4\,s + \frac{4\,s^2}{1-s},\, [4,\, 0,\, 0]]$$

```
> eta := CocycleOfExtension(M_,alpha1,M,alpha2,_M,R);
```

$$\eta := \begin{bmatrix} \dfrac{g}{c} + \dfrac{g\,\delta^2}{c} & 1 \end{bmatrix}$$

So one is able to compute η without computing neither $\mathrm{Hom}_R(K_1, M')$ nor $\mathrm{Ext}^1_R(M'', M')$. Actually, we only constructed η as an element of $\mathrm{Hom}_R(K_1, M')$, and one still needs to identify the cocycle represented by η inside $\mathrm{Ext}^1_R(M'', M')$ (recall, $\mathrm{Ext}^1_R(M'', M')$ is defined as the cokernel of $\mathrm{Hom}_R(P_0, M') \xrightarrow{d_1^*} \mathrm{Hom}_R(K_1, M')$, so it is a factor of $\mathrm{Hom}_R(K_1, M')$). For example, the cocycle represented by the morphism η might very well be trivial, even if η is non-trivial as a morphism from K_1 to M'. In case $\mathrm{Ext}^1_R(M'', M')$ is effectively computable (e.g. if R is commutative or M' is an R-bimodule), it is possible to ask `homalg` to return the element corresponding to η in $\mathrm{Ext}^1_R(M'', M')$, expressed (in a normalized way) in the abstract generators of the latter. A procedure to achieve this was assigned via calling `Ext` with the extra argument `"var_to_assign_proc_to_express_generators_abstractly"`, as done above:

```
> abstract_cocycle(eta);
```

$$[1,\, 0]$$

Again, in case $\mathrm{Hom}_R(K_1, M')$ and $\mathrm{Ext}^1_R(M'', M')$ are effectively computable, there is another way to obtain η as an element of $\mathrm{Ext}^1_R(M'', M')$. The short exact sequence $0 \to M' \to M \to M'' \to 0$ gives rise to the long exact Ext-sequence having $\mathrm{Hom}_R(M', M') \xrightarrow{\delta^0} \mathrm{Ext}^1_R(M'', M')$ as its first connecting homomorphism. It turns out that the extension cocycle η is the image of $\mathrm{id}_{M'}$ under δ^0. $\eta = \delta^0(\mathrm{id}_{M'})$ is called the *characteristic class* of the extension (cf. [CE99, below Prop. XI.9.2]). Following [CE99, Theorem XIV.1.1], the diagram

$$\begin{array}{ccccccc} & & F(M') & & & & \\ & & \downarrow{\scriptstyle F(\eta)} & \searrow^{\delta^0} & & & \\ F(P_0) & \xrightarrow[F(d_1)]{} & F(K_1) & \xrightarrow[\vartheta]{} & \mathrm{S}^1\, F(M') & \longrightarrow & 0 \end{array}$$

is commutative and gives (for $F = \mathrm{Hom}_R(-, M')$) the desired connecting homomorphism as the compositum $\delta^0 = F(\eta)\vartheta$.

Now identify the element `id` corresponding to $\mathrm{id}_{M'}$, expressed in the abstract generators of $\mathrm{Hom}_R(M', M')$:

```
>  hom :=
Hom(M_,M_,R,"var_to_assign_proc_to_express_generators_abstractly"='abs
tractly');
```

$$hom := \left[\left[[1, 0, 0, 0] = \begin{bmatrix} 0 & 0 \\ 1 & \frac{c}{g} \end{bmatrix}, [0, 1, 0, 0] = \begin{bmatrix} 0 & 1 \\ 0 & -\frac{g}{c} \end{bmatrix}, [0, 0, 1, 0] = \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}, [0, 0, 0, 1] = \begin{bmatrix} 0 & 0 \\ 0 & \delta^2 - 1 \end{bmatrix}\right], [[0, 0, 0, \mathrm{D}], [0, \mathrm{D}, \frac{\mathrm{D}\,g}{c}, 0], [\mathrm{D}, 0, 0, 0], [0, 0, \mathrm{D}\,\delta^2 - \mathrm{D}, 0]], \text{“Presentation”}, 4 + 9\,s + 15\,s^2 + s^3\,(\frac{15}{1-s} + \frac{6}{(1-s)^2}), [15, 6, 0]\right]$$

```
>  id := abstractly(matrix([[1, 0], [0, 1]]));
```

$$id := [0, 0, 1, 0]$$

The procedure `ch`, which was assigned above by `Ext`, computes for a given η the connecting homomorphism $\delta^0 := F(\eta)\vartheta$:

```
>  delta0:=ch(eta,M_);
```

$$\delta 0 := \begin{bmatrix} \frac{c}{g} & 1 \\ -\frac{g}{c} & 0 \\ 1 & 0 \\ -2 & 0 \end{bmatrix}$$

```
>  IsHom(hom,delta0,ext,R);
```

$$true$$

The image of $\mathrm{id}_{M'}$ under δ^0, expressed in the abstract generators of $\mathrm{Ext}^1_R(M'', M')$:

```
>  cc:=ImageOfElements(id,delta0,ext,R);
```

$$cc := [1, 0]$$

So we conclude that the extension $0 \to M' \xrightarrow{\alpha_1} M \xrightarrow{\alpha_2} M'' \to 0$ is non-split.

```
>  cc=OriginalElement(cc,ext,R);
```

$$[1, 0] = \begin{bmatrix} 0 & 1 \end{bmatrix}$$

And as explained above, the short exact sequence is reconstructable from η:

```
>  E2:=Extension(_M,eta,M_,R,"ALL");
```

$$E2 := [[[1,\,0]=[1,\,1,\,0],\,[0,\,1]=[0,\,-\frac{g\,(1+\delta^2)}{c},\,\mathrm{D}\,\delta]],\,[[\mathrm{D},\,\frac{c\,\mathrm{D}}{g}],\,[0,\,\mathrm{D}\,\delta^2-\mathrm{D}]],$$
$$\text{“Presentation”},\,2+5\,s+9\,s^2+s^3\,(\frac{9}{1-s}+\frac{4}{(1-s)^2}),\,[9,\,4,\,0]],$$
$$\left[\begin{array}{ccc} 0 & 0 & -1 \\ -\frac{g\,(1+\delta^2)}{c} & \mathrm{D}\,\delta & \frac{g\,(1+\delta^2)}{c} \end{array}\right],[$$
$$[[1,\,0,\,0]=[1,\,0,\,0,\,0],\,[0,\,1,\,0]=[0,\,1,\,0,\,0],\,[0,\,0,\,1]=[0,\,0,\,1,\,0]],$$
$$[[0,\,0,\,\mathrm{D}\,\delta^2-\mathrm{D}],\,[-\frac{\mathrm{D}\,\delta^2\,g}{c}-\frac{\mathrm{D}\,g}{c},\,\delta\,\mathrm{D}^2,\,\frac{\mathrm{D}\,g}{c}]],\ \text{“Presentation”},$$
$$3+9\,s+18\,s^2+s^3\,(\frac{18}{1-s}+\frac{9}{(1-s)^2}+\frac{1}{(1-s)^3}),\,[18,\,9,\,1]],\,\left[\begin{array}{cc} 1 & 0 \\ 0 & 1 \\ 0 & 0 \end{array}\right],[$$
$$[[1,\,0]=[1,\,0,\,0,\,0],\,[0,\,1]=[0,\,1,\,0,\,0]],\,[[-\frac{g\,\delta^2}{c}-\frac{g}{c},\,\mathrm{D}\,\delta]],\,\text{“Presentation”}\,,$$
$$2+6\,s+s^2\,(\frac{6}{1-s}+\frac{4}{(1-s)^2}+\frac{1}{(1-s)^3}),\,[6,\,4,\,1]]$$

```
>  IsShortExactSeq(E2,R,"VERBOSE");
```

$$\mathit{true}$$

Now, the delicate point is that from the above computations we don't get an explicit *equivalence* to the original M, although we know that one exists[2].

```
>  'M'=M; 'M2'=E2[3];
```

$$M=[[[1,\,0,\,0]=[1,\,0,\,0],\,[0,\,1,\,0]=[0,\,1,\,0],\,[0,\,0,\,1]=[0,\,0,\,1]],$$
$$[[\mathrm{D}\,\delta^2-\mathrm{D},\,\mathrm{D}\,\delta^2-\mathrm{D},\,0],\,[\frac{\mathrm{D}\,g}{c},\,-\frac{\mathrm{D}\,\delta^2\,g}{c},\,\delta\,\mathrm{D}^2]],\,\text{“Presentation”},$$
$$3+9\,s+18\,s^2+s^3\,(\frac{18}{1-s}+\frac{9}{(1-s)^2}+\frac{1}{(1-s)^3}),\,[18,\,9,\,1]]$$

$$M2=[[[1,\,0,\,0]=[1,\,0,\,0,\,0],\,[0,\,1,\,0]=[0,\,1,\,0,\,0],\,[0,\,0,\,1]=[0,\,0,\,1,\,0]],$$
$$[[0,\,0,\,\mathrm{D}\,\delta^2-\mathrm{D}],\,[-\frac{\mathrm{D}\,\delta^2\,g}{c}-\frac{\mathrm{D}\,g}{c},\,\delta\,\mathrm{D}^2,\,\frac{\mathrm{D}\,g}{c}]],\ \text{“Presentation”},$$
$$3+9\,s+18\,s^2+s^3\,(\frac{18}{1-s}+\frac{9}{(1-s)^2}+\frac{1}{(1-s)^3}),\,[18,\,9,\,1]]$$

In this particular simple example an equivalence can be quickly found:

```
>  zeta:=matrix([[-1, 0, 0], [1, 0, -1], [0, 1, 0]]);
```

$$\zeta := \left[\begin{array}{ccc} -1 & 0 & 0 \\ 1 & 0 & -1 \\ 0 & 1 & 0 \end{array}\right]$$

```
>  IsEquivalenceOfExtensions(E1,copy(zeta),E2,R,"VERBOSE");
```

$$\mathit{true}$$

Author: Mohamed Barakat and Barbara Bremer

[2]If the ring is commutative one can compute an isomorphism a posteriori (cf. [BR, Footnote 7 in Section 3.1.3]). For the noncommutative filtered situation one can proceed as in [QR05].

Date: 2007-07-30
Last modified: 2007-12-14 16:26

References

[BR] Mohamed Barakat and Daniel Robertz, homalg – *A meta-package for homological algebra*, submitted. (arXiv:math.AC/0701146 and http://wwwb.math.rwth-aachen.de/homalg). 3, 5

[BR08] ———, homalg *project*, 2003-2008, (http://wwwb.math.rwth-aachen.de/homalg).

[CE99] Henri Cartan and Samuel Eilenberg, *Homological algebra*, Princeton Landmarks in Mathematics, Princeton University Press, Princeton, NJ, 1999, With an appendix by David A. Buchsbaum, Reprint of the 1956 original. MR MR1731415 (2000h:18022) 2, 3

[HS97] P. J. Hilton and U. Stammbach, *A course in homological algebra*, second ed., Graduate Texts in Mathematics, vol. 4, Springer-Verlag, New York, 1997. MR MR1438546 (97k:18001) 2, 3

[QR05] Alban Quadrat and Daniel Robertz, *Parametrizing all solutions of uncontrollable multidimensional linear systems*, Proceedings of the 16th IFAC World Congress, Prague, Czech Republic, 2005. 5

[Rob06] D. Robertz, *Formal Computational Methods for Control Theory*, Ph.D. thesis, RWTH Aachen, Germany, June 2006, This thesis is available at http://darwin.bth.rwth-aachen.de/opus/volltexte/2006/1586.

[Rob08] Daniel Robertz, JanetOre*: A* Maple *package to compute a* Janet *basis for modules over* Ore *algebras*, 2003-2008, (http://wwwb.math.rwth-aachen.de:8040).

Lehrstuhl B für Mathematik, RWTH-Aachen University, 52062 Germany
E-mail address: mohamed.barakat@rwth-aachen.de

Lehrstuhl B für Mathematik, RWTH-Aachen University, 52062 Germany
E-mail address: barbara.bremer@rwth-aachen.de